
                          
INTRODUCTION

This application note discusses how to retrieve data
from program memory to data memory and write data
from data memory to program memory.

RETLW K Instruction

As in all PIC17CXXX family parts, the simplest method
used to retrieve data from program memory to data
memory is to use the RETLW K instruction. For
example:

; simple program to transfer

; table values to PortB

Main

movlw 5,W ;load offset

call SimpleTableRead

movwf PortB ;output to PortB

•

•

•     

SimpleTableRead

addwf PC ;add offset to PC

Table retlw 0 ;return a known

;table value based

;on the OFFSET.

•

•

•

retlw 10

In the example above, OFFSET is loaded with the
required offset to the Table and the subroutine
SimpleTableRead  is called. The table value is
returned in the W register. In this manner program
memory can be transferred to data memory.

Table Read Instruction

The PIC17C42 has an expanded instruction set which
includes the TABLRD and TLRD instructions. These
instructions are specifically constructed to transfer data
from program memory to data memory.

If the instruction syntax is: TABLRD t,i,f , the
sequence in which this instruction is executed is as
follows:

a) if t = 1 then the high byte of the table latch
(TBLATH) is loaded in the file register f.

b) else (if t = 0) the low byte of the table latch
(TBLATL) is loaded in the file register f.

c) next, the 16 bit data pointed to by the table
pointer (TBLPTR) is loaded into the table latch.

d) lastly, if i = 1 the table pointer (TBLPTR) is incre-
mented.

If the instruction syntax is: TLRD t,f , the sequence in
which this instruction is executed is as follows:

a) if t = 1 then the high byte of the table latch
(TBLATH) is loaded in the file register f.

b) else (if t = 0) the low byte of the table latch
(TBLATH) is loaded in the file register f.

FIGURE 1: TABLE READ

Note: The first time this instruction is executed in
a sequence, the table latch will not be
initialized, hence an unknown value will be
loaded in the file register. This is not a
problem if the user overwrites the same f
register in the next subsequent instruction.

Read In-Line

A simple method of transferring data from program
memory to data memory is to use the TABLRD and
TLRD instructions in sequence as shown in the exam-
ple below:

;transfer 6 bytes of data from program memory 
;at 0x500, to data memory at 0x80

ReadInLine

   movlw   05        ;load table pointer
                     ; with 0x500
   movwf   TBLPTRH   ;      /
   clrf    TBLPTRL   ;      /
   tablrd  0,1,0x80  ;get 16 bit value in
                     ;table latch.
   tlrd    0,0x80    ;low byte (1st) @ 80 
   tablrd  1,1,0x81  ;high byte (2nd) @ 81
   tlrd    0,0x82    ;3rd byte @ 82
   tablrd  1,1,0x83  ;4th byte @ 83
   tlrd    0,0x84    ;5th byte @ 84
   tablrd  1,1,0x85  ;6th byte @ 85

Reading a Block of Data

In instances where a block of N bytes needs to be trans-
ferred from program memory to data memory, the
TABLRD and TLRD instruction need to be included in a
loop which checks for N transfers.

;transfer 'COUNT' bytes (even values only) of 
;data at program memory 'MESSAGE' to data 
;memory at: 'RAM_BUFFER'
ReadBlock
   movlw   high MESSAGE   ; load table pointer
   movpf   W,TBLPTRH      ;      /
   movlw   low MESSAGE    ;      /
   movpf   W,TBLPTRL      ;      /
   bcf     ALUSTA,5       ;enable post auto
                          ; increment of FSR0
   movlw   RAM_BUFFER     ;initialize FSR0 
                          ;to RAM_BUFFER
   movfp   W,FSR0         ;        /
   movlw   COUNT/2        ;initialize count
   tablrd  1,1,RAM_BUFFER ;initialize table

                          ;latch

ReadBlockLoop
   tlrd    1,0x00         ;do indirect read
                          ;of high byte
   tablrd  0,1,0x00       ;do indirect read
                          ;of low byte
   decfsz  W              ;check if count=0
   goto    ReadBlockLoop  ;no then do next
   return                 ;else end of
                          ;transfer.
N = Number of bytes to transfer

Conclusion:

In cases where the number of bytes to be transferred is small, the Read In-Line offers small code size for fast transfer
rate. However, as the number of bytes to be transferred increases, the Read Block offers optimum code size for a decent
transfer rate.

Program Code Size Transfer Rate

Simple Table Read (using RETLW) N + 3 6 cycles/byte
Read In-Line 4 + N + N/2 1.5 cycles/byte
Read Block
(using loop)

14 + N/2 3 cycles/byte

Code Size

Simple Table Read Read In-Line Read Block

N = 10 13 19 19
N = 20 23 34 24
  1997 Microchip Technology Inc.

Table Write Instruction

The PIC17C42 has a TABLWT and a TLWT instruction
which transfer data from data memory to program
memory. Note in cases where the table pointer points to
internal EPROM, the table write instruction will try to
program the EPROM, hence the programming voltage
must be present on the VPP line to successfully
program the part.

The instruction syntax is: TABLWT t,i,f .

The sequence in which this instruction is executed is as
follows:

a) if t = 1 then the file register f is loaded to the high
byte of the table latch (TBLATH).

b) else (if t = 0) the file register f is loaded to the low
byte of the table latch (TBLATL).

c) next, the 16-bit data in the table latch is
transferred to the program memory pointed to by
the table pointer (TBLPTR).

d) lastly, if i = 1 the table pointer (TBLPTR) is
incremented.

The instruction syntax is: TLWT t,f

The sequence in which this instruction is executed is as
follows:

a) if t = 1 then the file register f is loaded to the high
byte of the table latch (TBLATH).

b) else (if t = 0) the file register f is loaded to the low
byte of the table latch (TBLATL).

FIGURE 2: TABLE WRITE

Write In-Line

A simple method of transferring data from data memory
to program memory is to use the TABLWT and TLWT
instruction in sequence as shown in the example below:

;transfer 6 bytes of data in data memory at

;0x80, to program memory at 0x5000:

ReadInLine

   movlw   50        ;load table pointer
                     ;with 0x5000
   movwf   TBLPTRH   ;      /
   clrf    TBLPTRL   ;      /
   tlwt    1,0x80    ; high byte @ table latch.
   tablwt  0,1,0x81  ; low byte @ table latch
                     ;latch  @ prog. mem.
   tlwt    1,0x82    ;3rd and 4th byte
                     ;@ prog. mem.
   tablwt  0,1,0x83  ;      /
   tlwt    1,0x84    ;5th and 6th byte
                     ;@ prog. mem.
   tablwt  0,1,0x85  ;      /   

Writing a Block of Data

In instances where a block of N bytes needs to be trans-
ferred from data memory to program memory, the
tablwt  and tlwt  instructions need to be included in a
loop which checks for N transfers.

;transfer 'COUNT' bytes (even values only) of 
;data at program memory at 'RAM_BUFFER' to 
;program memory at 'MESSAGE'

WriteBlock

   movlw   high MESSAGE   ; load table pointer
   movpf   W,TBLPTRH      ;      /
   movlw   low MESSAGE    ;      /
   movpf   W,TBLPTRL      ;      /
   bcf     ALUSTA,5       ;enable post auto
                          ; increment of FSR0
   movlw   RAM_BUFFER     ;initialize FSR0
                          ;to RAM_BUFFER
   movfp   W,FSR0         ;        /
   movlw   COUNT/2        ;initialize count
WriteBlockLoop
   tlwt    1,0x00         ;high byte
                          ;@ table latch
   tablwt  0,1,0x00       ;low byte @ table
                          ;latch;
                          ;table latch
                          ;@ prog. mem.
   decfsz  W              ; check if count = 0
   goto    WriteBlockLoop ;no then do next
   return                 ; else end of transfer
DS00548C-page 3

